

Getting started with the MegiQ VNA API in VB6


1 Introduction

This document shows how to create a new VNA project in VB6.

A VNA project must include a reference to the VNA API library so that the compiler can reference the entities in the library.

2 Creating a new Project

- Make sure the MegiQ VNA program is installed.
- Create a new project of the type you like (e.g. Forms Application).
- Go to menu 'Project | References...'



- Find MiQVNA in the list of Available references and select it.

3 Connecting to the VNA

Connecting to MiQVNA is as easy at this:

```
Private WithEvents clsVNA As mvnaVNAMain
Set clsVNA = New mvnaVNAMain
```

This will cause the MiQVNA program to be started and connected to your application. From this object all properties and methods as described in the MegiQ VNA Application Programming Interface can be used.

4 A skeleton program

The following is a template for a basic VNA application. Create a Form with a Command Button 'Button1' and a Label 'Label1' (default names). Copy this code into Form1.

```
Option Explicit

Private WithEvents clsVNA As mvnaVNAMain

Private Sub Form_Load()
    Set clsVNA = New mvnaVNAMain
    Call clsVNA_evtSystemStatus(clsVNA.SystemStatus)
End Sub

Private Sub Command1_Click()
    Call clsVNA.RunSweepOnce
End Sub

Private Sub clsVNA_evtDataChange(ByVal TraceNr As Long, ByVal NrTraces As Long)
    Dim TraceSet As mvnaTraceSet
    Dim IQData As mvnaIQData


    If (TraceNr = NrTraces) Then
        Set TraceSet = clsVNA.Measurement.TraceSet
        Set IQData = TraceSet.Traces(1).Channels("S11").DataSet("Return")

        ' Plot or handle IQData here

        ' For this demo, print 3 values
        Label1.Caption = "S11" & vbCrLf & _
            Format(IQData.PValue(0) / 1000000, "000") & "MHz: " & _
            Format(IQData.Value(0).AmpDB, "00.00") & "dB / " & _
            Format(IQData.Value(0).Phi180, "##0") & "deg" & vbCrLf & _
            Format(IQData.PValue(10) / 1000000, "000") & "MHz: " & _
            Format(IQData.Value(10).AmpDB, "00.00") & "dB / " & _
            Format(IQData.Value(10).Phi180, "##0") & "deg" & vbCrLf & _
            Format(IQData.PValue(20) / 1000000, "000") & "MHz: " & _
            Format(IQData.Value(20).AmpDB, "00.00") & "dB / " & _
            Format(IQData.Value(20).Phi180, "##0") & "deg" & vbCrLf
    End If
End Sub

Private Sub clsVNA_evtSystemStatus(ByVal Status As miqVNA.mvnaVNAStatus)
    If (Status = mvnaVST_Idle) Then
        Command1.Enabled = True
    Else
        Command1.Enabled = False
    End If
End Sub

Private Sub clsVNA_evtTerminate()
    Unload Me
End Sub
```



5 And further

You can find many functions and procedures for controlling the VNA, acquiring data and plotting results in the example VB6 project.